
ABSTRACT 
 

This paper displays an annotated SAS/GRAPH® plot of IBM z/OS® 

Goal Mode workload service class performance indexes by using 

macros and MXG® software.  The graph is generated via a batch job 

using the SAS® JCL procedure on an IBM mainframe server or Central 

Electronic Complex (CEC) using the z/OS platform.  The SAS Output 

Delivery System (ODS) is used to output the plot written as GIF file to 

a UNIX hierarchical file system HFS on the z/OS system.  The MXG 

software processes the IBM Resource Measurement Facility (RMF) 

workload activity record type seventy-two into a daily MXG computer performance database (PDB) 

dataset on z/OS Goal Mode workload performance data, i.e. the MXG PDB goal mode member 

PDB.TYPE72GO.  

 

The SAS/GRAPH plot generated from the PDB.TYPE72GO dataset is annotated with a large red star 

and text display at the performance index peak.  The z/OS workload service class, the time of the peak 

and the performance index value are displayed in the plot.  Displaying information in the SAS/GRAPH 

plot by using the annotate feature requires some additional SAS programming effort, but is very helpful 

when reviewing z/OS performance information for potential problems.  

 

KEYWORDS – z/OS  MXG  GRAPH  PLOT  ODS  ANNOTATE 

 

INTRODUCTION 

 

The IBM Resource Measurement Facility (RMF) record type seventy-two contains the detailed 

performance information on the z/OS goal mode workloads.  The z/OS workloads i.e. TSO users, batch 

jobs, database tasks etc, are composed of service classes.  An example would be having a BATCHT and 

BATCHP (i.e. test and production) service classes for the BATCH workload.  The service classes can 

also be broken down into multiple periods, i.e. periods 1, 2, 3 for short to long running tasks, so that a 

long running task in period three doesn’t excessively delay a short running HOTBATCH job. 

 

The RMF records are written to DASD as part of System Management Facility (SMF) processing, 

usually every 15 or 20 minutes or hourly, depending on each computer sites standards.  Shorter time 

intervals can be useful to research intermittent performance problems at the cost of creating more RMF 

records to process, while some sites use longer hourly intervals for system monitoring of overall 

performance or capacity. 

 

SAS/MXG software processes the z/OS SMF/RMF performance data into daily, weekly, monthly and 

trend MXG Computer Performance Databases. These databases contain numerous SAS datasets.  Each 

SAS dataset contains performance variables in a SAS readable format that can be processed by SAS 

jobs.  MXG also provides examples for analysis and plots using BASE SAS and SAS/GRAPH.  

PREPARATION 
 

SMF/RMF data is processed daily into an MXG Performance Database by MXG build processing.  The 

next step is to produce BASE SAS reports, charts, plots and optionally SAS/GRAPH charts and plots to 

review the daily computer performance information.  MXG provides MXGSASV8 and MXGSASV9 

JCL procedures for use.  The MXG supplied procedures provide excellent options, for production and 

testing new MXG software and processing larger amounts of SMF/RMF data.  However, this example 

uses the standard SAS proc that is shipped with SAS version 8 or version 9. 

 

This JCL example uses LIBRARY, SOURCLIB and PDB DD cards to reference the MXG FORMATS, 

SOURCLIB, and the daily performance database (PDB).  These files are used for most MXG 

processing. 

 

The daily SMF/RMF MXG processing builds the PDB, i.e. the z/OS dataset 

HLQ.MXG.PDB.DAILY(0).  HLQ is an abbreviation for high-level qualifier, the first qualifier of the 

dataset name.  The PDB is referenced in the JCL by the //PDB DD card.  The goal mode workload 

service class performance index information is in the PDB.TYPE72GO SAS dataset.  The TYPE72GO 

member is just one of numerous SAS datasets contained in the daily MXG Performance Database.   

 

Figure #1  SAS BATCH JOB JCL & USER SERVICE CLASS PERIOD PLOT MACRO   
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 //JOBNAME JOB  

 //******************************************************** 

 //* JCL FOR SAS GRAPH PERFORMANCE INDEX PLOT                 

 //******************************************************** 

 //STEP1    EXEC SASV913                                      

 //LIBRARY  DD DSN=HLQ.MXG.FORMATS,DISP=SHR                   

 //SOURCLIB DD DSN=HLQ.MXG.USER.SOURCLIB,DISP=SHR             

 //         DD DSN=HLQ.MXG.SOURCLIB,DISP=SHR                  

 //PDB      DD DSN=HLQ.MXG.PDB.DAILY(0),DISP=SHR              

 //SYSIN    DD * 

                                                                   

 %INCLUDE SOURCLIB (USERMACS);  /* Include my user macros     */   

 %LET PLOTDAY = DAY(TODAY()-1); /* PLOT DAY USUALLY YESTERDAY */   

                                                                   

 DATA _NULL_;  /* USE A DIFFERENT PATH FOR EACH DAY OF MONTH  */   

 CALL SYMPUT('USERPATH','/u/mxg/' || PUT(&PLOTDAY,Z2.0)  );        

 RUN;                                                              

                                                                   

 %USER_SERVICE_CLASS_PERIOD_PLOT (  

    PLOT_TITLE = z/OS Computer Server ,      /* PLOT TITLE1   */   

    PDB = PDB,                               /* PDB  DD CARD  */   

    PLOTVAR=PERFINDX,        /* PERFORMANCE VARIABLE TO PLOT  */   

    PATH=&USERPATH,          /* PATH TO WRITE THE PLOT TO     */   

                             /* i.e. /u/mxg/01 to ...31       */   

    SYSPLEX_TO_PLOT=SUGPLEX, /* COMPUTER SYSPLEX TO SELECT    */   

    SYSTEM_TO_PLOT=SYS8,     /* COMPUTER SYSTEM ID TO SELECT  */   

    DAY_TO_PLOT = &PLOTDAY   /* DAY TO PLOT I.E. 1,2,3...31   */   
     );      

SERVICE CLASS PERFORMANCE INDEX DETAILS 
 

The MXG PDB.TYPE72GO dataset contains over two hundred performance variables and thousands of 

observations.  MXG sorts the data by computer sysplex, computer system and more.  The TYPE72GO 

data observations include timestamps with the date and time the information was collected and written 

to DASD, the duration of the time interval and the CPU model number. 

 

The performance index for each service class is calculated for each period in the service class.  Thus, a 

daily SAS/GRAPH plot of PERFINDX against TIME should plot the variable for each service class 

period to be useful.  This example concatenates the MXG SRVCLASS and PERIOD variables to create 

the plot classification variable.  The TRIM function is used to remove any trailing blanks from the 

service class. 

 

Figure #2  CREATE AND LABEL THE PLOT CLASSIFICATION VARIABLE 

 

 

 

 

 

 

 
 

Notice that the label for the PLOT_CLASS variable is displayed in the bottom center of the plot legend 

in Figure #3.  Many service classes only have one service class period, i.e. period 1.  However, some 

workloads have multiple service class periods i.e. periods 1, 2, 3 to give priority and permit short 

running work or TSO transaction to complete without being delayed for hours by a long running task. 

 

The SAS/GRAPH plot of the PERFINDX variable is useful to review z/OS workload service classes 

meeting or not meeting goals.  The PERFINDX variable is a ratio of several z/OS performance metrics.  

The basic PERFINDX example for an average response time goal would be calculated as the actual 

response time divided by the response goal of the workload.  Depending on how the service class period 

goals are defined can complicate the above explanation and calculations. 
 

Z/OS technical performance specialists consider a workload with a PERFINDX of exactly 1.0 as 

meeting the assigned goal.  They also consider a workload with a performance index value above 

1.0 as not meeting its goal, and it may be experiencing delays, or may have an ambitiously 

exuberant goal that should be reviewed.  Its usually ok for workloads to exceed goals, i.e. a 

performance index value under 1.0, but that may or may not be a problem if a more important 

workload is missing its goal.  Thus reviewing the performance index is usually an ongoing task 

of z/OS computer performance monitoring. 

 

USER MACRO PROGRAM DETAILS 
 

The program for this example is a macro with several input parameters.  Since it is a user macro, i.e. a 

macro written by a user i.e. myself, the code is in my user source library referenced by the JCL.  My 

user macros are in my z/OS source library member USERMACS.  And as seen in FIGURE 1 the code 

uses an %INCLUDE statement to include this source member.   
 

%INCLUDE SOURCLIB (USERMACS); 

 

 LABEL PLOT_CLASS=’SERVICE CLASS AND PERIOD’; 

 

 /* CONCATENATE THE SERVICE CLASS AND PERIOD FOR THE PLOT      */ 

 PLOT_CLASS = TRIM (SRVCLASS) || '_' || PUT(PERIOD,Z1.0); 

Figure 1 also shows the input parameters to the user macro.  The parameters are keyword named 

parameters i.e. the parameters do not need to be listed in a specific positional order.  A small amount of 

SAS code is used before invoking the user macro. 

 

The %LET statement is used create the &PLOTDAY variable and to select the day to plot as the day of 

the month for yesterday, i.e. DAY (TODAY() –1).  Since the day to be plotted is usually yesterday. 

 

The DATA _NULL_ data step is used to create the &USERPATH variable that is input to the plot 

macro for its PATH parameter.  The CALL SYMPUT function is used here to assign a path or value to 

the variable USERPATH and to concatenate the day of the month i.e. 01, 02, 03…31 to the path. 

 

PLOT_TITLE.  This is a self-explaining parameter for input to the macro.  The title information will 

be displayed in the web page title and the first line of the SAS/GRAPH plot title.   

 

PATH.  The plot is written to a UNIX hierarchical file system HFS file on the z/OS system.  This 

example uses the previous &USERPATH variable that was created from the prior SYMPUT function.  

The usual path for initial testing is usually /u/tsouserid.  Concatenating the day of the month, i.e. 01, 02, 

03…29, 30, 31 to the path allows for automated plots to be written and reviewed daily. 

 

PLOTVAR.  This is the performance variable to plot vs. time of day.  The MXG variable PERFINDX 

is selected for this example.  The advantage of using a SAS macro is that additional performance 

variables can be reviewed and plotted by using the macro again by just changing the PLOTVAR.  The 

name of the plot will be &PLOTVAR.gif and the web page body will be 

SERVICECLASS_&PLOTVAR..HTML. The plot variable will also be displayed in the title and 

footnote. 

 

PDB.  This parameter is for the DDNAME on the //DDNAME DD card referencing the MXG 

performance database that is used for input.  The parameter used is almost always PDB, but might vary 

to PDBWEEK or PDBMONTH for ad hoc reports. 

 

SYSPLEX_TO_PLOT.  Some computer sites may have many computer systems forming multiple 

sysplex’s.  This macro variable can be used to limit the plots to one sysplex.  

 

SYSTEM_TO_PLOT.  Some computer sites load the RMF/SMF data into one large MXG PDB and 

multiple computer systems can be in the TYPE72GO dataset.  Specifying the SYSTEM_TO_PLOT that 

is identified in the z/OS SMFPARM00 member can be used to limit the plot to one system id.  Sites with 

duplicate ‘SYSTEM’ members can modify this user macro to additionally select/deselect the 

‘SYSNAME’ identifier from z/OS IEASYS00. 

 

DAY_TO_PLOT.  For this example this parameter is set to &PLOTDAY from the %LET statement, 

which is usually set to yesterday by using the DAY function, i.e. DAY (TODAY()-1).  This helps in 

selecting the day for the plot and filtering the data, to only one twenty-four hour day for each plot. 

 

RTRANS_MIN.  This is set to a minimum transaction count for percent response transactions during 

the RMF monitoring time duration interval.  This is to delete observations with very low transaction 

counts i.e. low activity.  Since, the PERFINDX values can sometimes become skewed and distort the 

plot when there is very low activity or if the workload tasks are in a 99% idle state.  Users can modify or 

add additional user macro parameters to edit or filter the input data as needed. 

Figure  #3  Annotated SAS/GRAPH®  Plot of z/OS Goal Mode Performance Indexes  
  

    

 
 

 

SCREEN PRINT OF THE ANNOTATED SAS/GRAPH PLOT 
 

 

 

Figure #3 displays a print screen of the SAS/GRAPH z/OS Goal Mode performance indexes for a 

simulated day.  The plot was generated using the Output Delivery System (ODS) to send the output to a 

HFS file.  Notice that the web page title and the SAS/GRAPH plot title use input from the 

“&PLOT_TITLE” user macro variable.  A modified test MXG® PDB was used for input.  

 

A large red star and text annotate the peak at 2.4 for the simulated BATCH SAS workload in service 

period 1 at 18:30.  The remaining performance index values mostly range from just above and below the 

yellow reference line at 1.0.  To provide more information, the values for the plot byvals i.e. computer 

sysplex, system, sysname and date are included in the plot title.   

 

The peaks can fluctuate daily by workload and time of the peak.  Reviewing the PERFINDX plot for 

elevated values and peaks can be useful to detect performance and response problems before users 

complain about response time delays.  Performance tuning actions could include raising the workloads 

goal, lowering other z/OS workload goals, software tuning changes within the workload, z/OS system 

changes, and reviewing overall CPU usage and DASD I/O activity for delays. 

ANNOTATE DATASET FOR SAS/GRAPH® PROC GPLOT      
 

Figure #4 Assign variable values for the 1st observation of the Annotate Dataset 
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

To highlight and annotate the peak &PLOTVAR variable value, the peak value must be found.  Thus 

some additional processing is needed to find the maximum value, the time of the peak and the service 

class of the peak.  The maximum value or peak information data (see appendix for detail) is input to the 

annotate dataset. 

 

Figure #4 shows the DATA step to create the 1
st
 observation of the annotate dataset, which places a red 

star above the desired plot (x, y) coordinate location.  The annotate dataset uses information from the 

maximum values dataset for the values of the plot coordinates.  The annotate dataset for this plot will 

have three observations.  The second observation (see appendix) contains the plot value and service 

class to be displayed at an angle.  The third observation contains text to display the time of the 

maximum. 

 

Figure #5.  REFERENCING THE ANNOTATE DATASET  
 

 

 

 

 

 

 
 

As seen in Figure #5, the annotate option is used in the PLOT statement of the SAS/GRAPH GPLOT 

procedure.  The annotate option informs the procedure that the annotating feature is to be used and 

identifies the SAS dataset that contains the annotation functions.  

 DATA USER_ANNOTATE_DATASET; 

 

 /* DEFINE LENGTH OF ANNOTATE CHARACTER VARIABLES               */  

 LENGTH FUNCTION STYLE COLOR $8    TEXT $20;                        

                                                                    

 /* ANNOTATE VARIABLES AND VALUES FOR COORDINATES AND TEXT SIZE */  

 /* ANNOTATE VARIABLE WHEN TO ANNOTATE i.e. AFTER PLOT IS MADE  */  

 RETAIN XSYS '2' YSYS '2' HSYS '1' WHEN 'a';                        

                                                                    

 SET USER_MAX_VALUES;         /* INPUT IS THE MAX VALUE DATASET */  

   X=TIME;                    /* X ANNOTATE PLOT COORDINATE     */  

   Y=&PLOTVAR;                /* Y ANNOTATE PLOT COORDINATE     */  

                                                                    

                       /* PLACE RED STAR ABOVE X, Y COORDINATES */  

   FUNCTION='LABEL';          /* ANNOTATE TASK WHAT TO DO       */  

   COLOR='RED';               /* ANNOTATE COLOR INFORMATION     */  

   SIZE=10;                   /* ANNOTATE TEXT SIZE INFORMATION */  

   POSITION='b';              /* POSITIONED CENTER ABOVE X,Y    */  

   STYLE='SPECIAL';           /* SELECT SPECIAL STYLE           */  

   TEXT='M';                  /* SPECIAL STYLE STAR CHARACTER   */  

   OUTPUT;                    /* OUTPUT THE 1st OBSERVATION     */  

  PLOT &PLOTVAR*TIME=PLOT_CLASS / 

      NAME="&PLOTVAR"                     /* NAME plotname.gif */  

      CFRAME=CXEEFFFF                     /* COLOR OF FRAME    */  

      ANNOTATE=USER_ANNOTATE_DATASET      /* ANNOTATE DATA     */  

CONCLUSION 

 
The z/OS Base SAS JCL procedure is used in this example, and daily jobs are scheduled by automated 

software.  Using SAS Output Delivery System (ODS) and SAS/GRAPH to create displays of z/OS 

performance plots results in high quality graphical output that is easily reviewed.  

 

The “USER_SERVICE_CLASS_PERIOD_PLOT” macro can be changed for larger or smaller 

computer sites needs.  But the advantage of the macro is that it can be invoked over and over again with 

different plot variables to review several performance metrics without writing additional SAS code. 

 

Annotating the SAS/GRAPH plot of z/OS service class performance indexes takes some additional 

programming effort.  But the result provides an easy to read display of useful information and has a 

positive visual impact.  Reviewing z/OS performance information is also simplified and streamlined by 

having the peak service class performance index value identified, highlighted and time stamped. 

APPENDIX – MACRO CODE. 

 
Text can be copied from the PDF document by using the ‘select text tool’ icon on the PDF toolbar. 
 

 %MACRO USER_MXG_GIF_GOPTIONS;      /* GRAPHIC OPTIONS FOR GRAPHS */ 
     GOPTIONS RESET=GLOBAL          /* RESET OPTIONS              */ 

     DEVICE=GIF                     /* GIF DEVICE FOR ODS GRAPHS  */ 

     GUNIT=PCT                      /* GRAPHICS PCT UNITS         */ 

     XMAX=10.0IN YMAX=6.0IN         /* XMAX YMAX GRAPHIC VALUES   */ 
     CBACK=CXFFFFFE                 /* COLOR BACKGROUND           */ 

     FTEXT=SIMPLEX                  /* FONT TEXT                  */ 

     CTEXT=BLUE                     /* COLOR TEXT SELECTED        */ 

     HTEXT=2.2 ;                    /* HEIGHT OF TEXT             */ 
 

     /* GRAPH PLOT SYMBOLS – COLOR LINE TYPE WIDTH VALUE HEIGHT   */ 

     SYMBOL1  INTERPOL=JOIN C=GREEN    L=1 W=2 V=NONE; 

     SYMBOL2  INTERPOL=JOIN C=RED      L=1 W=2 V=NONE; 

     SYMBOL3  INTERPOL=JOIN C=CYAN     L=1 W=2 V=NONE; 

     SYMBOL4  INTERPOL=JOIN C=BLACK    L=1 W=2 V=NONE; 

     SYMBOL5  INTERPOL=JOIN C=BLUE     L=1 W=2 V=NONE; 

     SYMBOL6  INTERPOL=JOIN C=MAGENTA  L=2 W=2 V=NONE; 

     SYMBOL7  INTERPOL=JOIN C=BROWN    L=2 W=2 V=NONE; 

     SYMBOL8  INTERPOL=JOIN C=ORANGE   L=2 W=2 V=NONE; 

     SYMBOL9  INTERPOL=JOIN C=PURPLE   L=2 W=2 V=NONE; 

     SYMBOL10 INTERPOL=JOIN C=GREY     L=2 W=2 V=NONE; 
     /* up to 255 symbol statements                               */ 

     SYMBOL20 INTERPOL=JOIN C=GREY     L=2 W=3 V=SQUARE   H=4; 

 %MEND USER_MXG_GIF_GOPTIONS;       /* END OF THIS MACRO          */ 
 

 

 %MACRO USER_SERVICE_CLASS_PERIOD_PLOT 
  (PLOT_TITLE = User Text Title, 

   PDB = PDB, 

   PLOTVAR=PERFINDX, 

   PATH="/u/mxg/01/", 

   DAY_TO_PLOT = DAY(TODAY()-1), 
   SYSPLEX_TO_PLOT =, 

   SYSTEM_TO_PLOT  =, 

   RTRANS_MIN      =15 
   ); 

 

 /* SAS GRAPH PLOT OF GOAL MODE PERFORMANCE VARIABLES              */ 

 /* INPUT PARAMETERS                                               */ 

 /*   PLOT_TITLE       TITLE FOR THE SAS GRAPH PLOT AND WEB PAGE   */ 

 /*   PDB              PDB DD CARD i.e. USUALLY PDB OR PDBWEEK     */ 

 /*   PLOTVAR          PERFORMANCE VARIABLE TO PLOT                */ 

 /*   PATH             PATH TO WRITE THE PLOT TO                   */ 

 /*   DAY_TO_PLOT      DAYNUMBER OF MONTH   I.E. 1,2,3...31        */ 

 /*   SYSPLEX_TO_PLOT  SYSPLEX TO SELECT                           */ 

 /*   SYSTEM_TO_PLOT   SYSTEM  TO SELECT                           */ 

 /*   RTRANS_MIN       MINIMUM TRANS COUNT FOR PCT RESPONSE GOAL   */ 

 /*                    I.E. VERY LOW COUNTS SKEW PERFINDX          */ 

OPTIONS  NODATE NONUMBER NOBYLINE ; 

 

DATA USER_GOALMODE_DATA; 

   FORMAT PLOT_CLASS $CHAR10. DATE WEEKDATE29. TIME TIME5.0 ; 

   LABEL  PLOT_CLASS='SERVICE CLASS AND PERIOD' 

          TIME='TIME*OF*DAY'; 

   KEEP   SYSPLEX SYSTEM SYSNAME DATE TIME &PLOTVAR PLOT_CLASS; 

 

   SET &PDB..TYPE72GO ;                 /* READ THE INPUT DATASET */ 
        /* MACRO FILTER - IF SYSTEM SYSPLEX TO PLOT SPECIFIED     */ 

        %IF %LENGTH(&SYSTEM_TO_PLOT)  > 0  %THEN %DO; 
         IF SYSTEM NE "&SYSTEM_TO_PLOT"  THEN DELETE ;   %END; 

 

        %IF %LENGTH(&SYSPLEX_TO_PLOT)  > 0  %THEN %DO; 
         IF SYSPLEX NE "&SYSPLEX_TO_PLOT" THEN DELETE ;  %END; 

 

        DATE=DATEPART(STARTIME); 

 

        %IF %LENGTH(&DAY_TO_PLOT)  > 0   %THEN %DO; 
         IF DAY(DATE)    NE &DAY_TO_PLOT   THEN DELETE ; %END; 

 

        TIME=TIMEPART(STARTIME); 

        /* OPTIONAL NOT INCLUDING REPORT CLASSES IN THE PLOT      */ 

        IF RPRTCLAS = 'Y' THEN DELETE; 

 

        /* DELETE MISSING VALUES AND ZERO VALUES                  */ 

        IF &PLOTVAR IN ( . , 0)  THEN DELETE;  

 

        /* percent response time transactions delete low activity */ 

        IF ((R723CRGF ='P') AND (TRANS LE &RTRANS_MIN) ) THEN DELETE; 

 

        /* CONCATENATE THE SERVICE CLASS AND PERIOD FOR THE PLOT  */ 

        /* AND TRIM ANY SERVICE CLASS TRAILING BLANKS             */ 

        PLOT_CLASS = TRIM (SRVCLASS) || '_' || PUT(PERIOD,Z1.0); 

 

 RUN;   /* EXECUTE THE DATA STEP                                  */  

 

 /* FOR ANNOTATION BY SORTED BY VALUES                            */  

 /* PREPARE TO FIND THE PLOT VARIABLE MAXIMUM, TIME OF THE MAXIMUM*/  

 /* AND PLOT CLASS WITH THE SERVICE CLASS AND PERIOD              */  

 

PROC SORT DATA=USER_GOALMODE_DATA                                   

      OUT=USER_MAX_VALUES;                                          

      BY SYSPLEX SYSTEM SYSNAME DATE DESCENDING &PLOTVAR;           

RUN;                                                                

                                                                    

DATA USER_MAX_VALUES;                                               

 /* SELECT FIRST BY VALUE OBSERVATION(S) I.E. PLOT VAR MAXIMUM    */  

 /* TIME OF THE MAXIMUM AND SERVICE CLASS INFORMATION             */  

  SET USER_MAX_VALUES; BY SYSPLEX SYSTEM SYSNAME DATE;            

    IF FIRST.SYSPLEX OR FIRST.SYSTEM OR                           

       FIRST.SYSNAME OR FIRST.DATE;                               

RUN; 

 DATA USER_ANNOTATE_DATASET; 

   /* DEFINE THE LENGTH OF ANNOTATE CHARACTER VARIABLES           */ 

   LENGTH  FUNCTION STYLE COLOR $8    TEXT $20; 
   /* ANNOTATE VARIABLES VALUES FOR COORDINATES AND TEXT HEIGHT   */ 

   /* ANNOTATE VARIABLE WHEN TO ANNOTATE i.e. AFTER PLOT IS MADE  */ 

   RETAIN  XSYS '2' YSYS '2' HSYS '1' WHEN 'a'; 
 

   SET USER_MAX_VALUES;         /* INPUT IS THE MAX VALUE DATASET */ 

     X=TIME;                    /* X AND Y ANNOTATION VARIABLES   */ 

     Y=&PLOTVAR;                /* FOR ANNOTATION PLOT COORDINATES*/ 
 

   /* PLACE RED STAR ABOVE X, Y COORDINATES                       */ 

     FUNCTION='LABEL';          /* ANNOTATE FUNCTION WHAT TO DO   */ 

     COLOR='RED';               /* COLOR INFORMATION              */ 

     SIZE=10;                   /* SIZE OF TEXT OR ITEM           */ 
     STYLE='SPECIAL';           /* SELECT SPECIAL STYLE           */ 

     POSITION='b';              /* POSITIONED CENTER ABOVE X, Y   */ 

     TEXT='M';                  /* SPECIAL STYLE STAR CHARACTER   */ 

     OUTPUT;                    /* OUTPUT THE OBSERVATION         */ 
 

   /* ADD ANNOTATED TEXT.  PLOT VALUE, SERVICE CLASS AND PERIOD   */ 

     ANGLE=5;                   /* ANGLE TO DRAW                  */ 
     COLOR='PURPLE';            /* COLOR TO DRAW TEXT             */ 

     SIZE=5;                    /* SIZE OF TEXT                   */ 
     STYLE='SWISSBI';           /* TEXT STYLE                     */ 

     POSITION='3';              /* LEFT ALIGNED CELL ABOVE X, Y   */ 

     TEXT= '  ' || LEFT(PUT(Y,6.1)) || PUT(PLOT_CLASS,$char10.);     
     OUTPUT;                    /* OUTPUT THE OBSERVATION         */ 
 

   /* ADDITIONAL ANNOTATED TEXT LINE.  TIME OF THE PLOT MAXIMUM   */ 

     POSITION='6';              /* LEFT ALIGNED CELL AT X, Y      */ 

     TEXT = '  at ' || PUT(TIME,TIME5.0); 

     OUTPUT;                    /* OUTPUT THE OBSERVATION         */ 

 RUN; 
 

  /* PRINT LISTING OF THE ANNOTATE DATA FOR INFORMATIONAL REVIEW  */ 

 PROC PRINT DATA=USER_ANNOTATE_DATASET;  

   BY SYSPLEX SYSTEM SYSNAME DATE; 

     TITLE1 "&PLOT_TITLE Annotation data for &PLOTVAR"; 

     TITLE2 '#BYVAR1= #BYVAL1  #BYVAR2= #BYVAL2  #BYVAR3= #BYVAL3'; 

     TITLE3 '#BYVAL4'; 

     FOOTNOTE1;                 /* CLEAR THE FOOTNOTES            */ 

 RUN;                           /* RUN PROC PRINT                 */ 
 

 ODS LISTING CLOSE;             /* CLOSE LISTING DESTINATION      */ 

  

 ODS HTML PATH=”&PATH” (URL=NONE)               /* WEB PATH INFO  */ 

          BODY="SERVICECLASS_&PLOTVAR..HTML"    /* WEB PAGE BODY  */ 

          (TITLE="&PLOT_TITLE FOR &PLOTVAR ");  /* WEB PAGE TITLE */ 
 

 %USER_MXG_GIF_GOPTIONS;        /* SELECT MY USER GOPTIONS MACRO  */ 

 RUN; 

 PROC GPLOT DATA=USER_GOALMODE_DATA;       /* THE GPLOT PROCEDURE*/ 

   BY SYSPLEX SYSTEM SYSNAME DATE; 

 

     TITLE1 HEIGHT=3.0 C=BLUE "&PLOT_TITLE Plot of"  
                       C=RED  " &PLOTVAR"; 

 

     TITLE2 HEIGHT=3.0 C=BLUE  '#BYVAR1= ' C=RED '#BYVAL1'  
                       C=BLUE ' #BYVAR2= ' C=RED '#BYVAL2'  

                       C=BLUE ' #BYVAR3= ' C=RED '#BYVAL3'; 

 

     TITLE3 HEIGHT=2.8 C=VIRO '#BYVAL4'; 
 

     FOOTNOTE1 H=2.2 J=L ' SOURCE: MXG PERFORMANCE DATABASE' 
                     J=R 'SAS/GRAPH ' C=RED "&PLOTVAR PLOT "; 

     FOOTNOTE2 H=1 ' ';                    /* SPACE FOR BORDER   */ 
 

     AXIS1  COLOR=BLACK                    /* VERITICAL AXIS     */ 

            LABEL=(HEIGHT=2.2 COLOR=RED ROTATE=90 ANGLE=-90) 

            MAJOR=(HEIGHT=2.0 COLOR=GREEN) /* MAJOR TICKS        */ 

            MINOR=(H=1 N=4 COLOR=RED)      /* MINOR TICKS        */ 

            VALUE=(H=3.0 COLOR=PURPLE)     /* AXIS VALUES        */ 
            ; 

     AXIS2  COLOR=BLACK                   /* 24hr HORIZONTAL AXIS*/ 

            LABEL=(HEIGHT=2.2 C=BLACK)    /* LABEL HEIGHT        */ 

            ORDER=('00:00'T TO '24:00'T  BY '02:00'T) 

            MAJOR=(HEIGHT=2 COLOR=RED)    /* MAJOR TICK MARKS    */ 

            MINOR=(H=1 N=1 C=BIB)         /* MINOR TICK MARKS    */ 

            VALUE=(H=2.0 COLOR=BROWN)     /* AXIS VALUES         */ 
            ; 

     LEGEND1 VALUE=(HEIGHT=2.0)           /* LEGEND TEXT HEIGHT  */ 

            ACROSS=5                      /* ROW 5 ITEMS WIDE    */ 
            CFRAME=WHITE                  /* LEGEND FRAME COLOR  */ 

            CBORDER=GREEN                 /* LEGEND BORDER COLOR */ 

            LABEL=(H=2.5 C=MAGENTA POSITION=(BOTTOM CENTER)) ; 
 

     PLOT  &PLOTVAR*TIME=PLOT_CLASS / 

           NAME="&PLOTVAR"                 /* name plotname.gif  */ 

           CFRAME=CXEEFFFF                 /* PLOT FRAME COLOR   */ 

           ANNOTATE=USER_ANNOTATE_DATASET  /* ANNOTATE DATASET   */ 

           VAXIS=AXIS1                     /* VERTICAL AXIS      */ 

           VZERO                           /* START AXIS AT ZERO */ 

           VREF=(.97 to 1.03 by .01)       /* LOTS OF VREF LINES */ 
           CVREF='YELLOW'                  /* VREF LINE COLOR    */ 

           HAXIS=AXIS2                     /* HORIZONTAL AXIS    */ 

           LEGEND=LEGEND1;                 /* LEGEND             */ 

   RUN; QUIT;                              /* RUN AND CLOSE GPLOT*/ 

   ODS HTML CLOSE;                         /* CLOSE HTML         */ 

   ODS LISTING;                            /* LISTING RESET OPEN */ 

   RUN; 

 

 %MEND USER_SERVICE_CLASS_PERIOD_PLOT;     /* END OF THE MACRO   */ 

